

# NAG Library Routine Document

## F07AWF (ZGETRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07AWF (ZGETRI) computes the inverse of a complex matrix  $A$ , where  $A$  has been factorized by F07ARF (ZGETRF).

### 2 Specification

```
SUBROUTINE F07AWF (N, A, LDA, IPIV, WORK, LWORK, INFO)
  INTEGER          N, LDA, IPIV(*), LWORK, INFO
  COMPLEX (KIND=nag_wp) A(LDA,*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *zgetri*.

### 3 Description

F07AWF (ZGETRI) is used to compute the inverse of a complex matrix  $A$ , the routine must be preceded by a call to F07ARF (ZGETRF), which computes the  $LU$  factorization of  $A$  as  $A = PLU$ . The inverse of  $A$  is computed by forming  $U^{-1}$  and then solving the equation  $XPL = U^{-1}$  for  $X$ .

### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $LU$  factorization of  $A$ , as returned by F07ARF (ZGETRF).  
*On exit:* the factorization is overwritten by the  $n$  by  $n$  matrix  $A^{-1}$ .
- 3: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07AWF (ZGETRI) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 4: IPIV(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On entry:* the pivot indices, as returned by F07ARF (ZGETRF).

- 5: WORK(max(1,LWORK)) – COMPLEX (KIND=nag\_wp) array *Workspace*  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 6: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F07AWF (ZGETRI) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).  
*Suggested value:* for optimum performance LWORK should be at least  $N \times nb$ , where *nb* is the **block size**.  
*Constraint:* LWORK  $\geq$  max(1, N) or LWORK = -1.
- 7: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO = -*i*, argument *i* had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle value \rangle$  of the diagonal is zero. *U* is singular, and the inverse of *A* cannot be computed.

## 7 Accuracy

The computed inverse *X* satisfies a bound of the form:

$$|XA - I| \leq c(n)\epsilon|X|P|L||U|,$$

where  $c(n)$  is a modest linear function of *n*, and  $\epsilon$  is the **machine precision**.

Note that a similar bound for  $|AX - I|$  cannot be guaranteed, although it is almost always satisfied. See Du Croz and Higham (1992).

## 8 Parallelism and Performance

F07AWF (ZGETRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $\frac{16}{3}n^3$ .

The real analogue of this routine is F07AJF (DGETRI).

## 10 Example

This example computes the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix}.$$

Here  $A$  is nonsymmetric and must first be factorized by F07ARF (ZGETRF).

### 10.1 Program Text

```

Program f07awfe

!      F07AWF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: nag_wp, x04dbf, zgetrf, zgetri
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: i, ifail, info, lda, lwork, n
!      .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), work(:)
Integer, Allocatable        :: ipiv(:)
Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
Write (nout,*) 'F07AWF Example Program Results'
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
lwork = 64*n
Allocate (a(lda,n),work(lwork),ipiv(n))

!      Read A from data file

Read (nin,*)(a(i,1:n),i=1,n)

!      Factorize A
!      The NAG name equivalent of zgetrf is f07arf
Call zgetrf(n,n,a,lda,ipiv,info)

Write (nout,*)
Flush (nout)
If (info==0) Then

!      Compute inverse of A
!      The NAG name equivalent of zgetri is f07awf
Call zgetri(n,a,lda,ipiv,work,lwork,info)

!      Print inverse

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04dbf('General',' ',n,n,a,lda,'Bracketed','F7.4','Inverse',      &
           'Integer',rlabs,'Integer',clabs,80,0,ifail)

Else
Write (nout,*) 'The factor U is singular'
End If

End Program f07awfe

```

## 10.2 Program Data

F07AWF Example Program Data

```

4
(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92) :Value of N
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

```

## 10.3 Program Results

F07AWF Example Program Results

```

Inverse
1 2 3 4
1 ( 0.0757,-0.4324) ( 1.6512,-3.1342) ( 1.2663, 0.0418) ( 3.8181, 1.1195)
2 (-0.1942, 0.0798) (-1.1900,-0.1426) (-0.2401,-0.5889) (-0.0101,-1.4969)
3 (-0.0957,-0.0491) ( 0.7371,-0.4290) ( 0.3224, 0.0776) ( 0.6887, 0.7891)
4 ( 0.3702,-0.5040) ( 3.7253,-3.1813) ( 1.7014, 0.7267) ( 3.9367, 3.3255)

```

---